
1. Introduction

This document provides basic information on using the TIROS-3 Medium-Resolution Scanning 

Radiometer Level 1 Final Meteorological Radiation Data (or FMRT) product.

1.1 Data Product Description

The TIROS 3 Level-1 Medium-Resolution Scanning Radiometer Final Meteorological Radiation Data 

(FMRT) product contains radiances measured in 5 infrared wavelength regions, expressed in either 

equivalent blackbody temperature (channels 1, 2 and 4) or effective radiant emmitance (channels 3 

and 5). The data were originally created on IBM 7090 computers and eventually copied to 6250 bpi 

7-track magnetic tapes (originally stored on 200 bpi 7-track tapes) . The data from these tapes were 

recovered from the magnetic tapes and are now archived in digital files in their original binary file 

format. The data are grouped into swaths broken into multiple blocks (typically 10-20) with 5 

measurements for each of the five channels. Geolocation, time and other ancillary information are 

included. A swath will trace an elliptical, parabolic, or hyperbolic pattern on the ground due to the 

rotating of the instrument about the satellite spin axis.

The data are available for the time period from 12 July 1961 to 30 September 1961. Each data file 

contains a single orbit of data. The principal investigators for the TIROS 3 scanning radiometer 

experiment was Joseph D. Barksdale from NASA/GSFC.

This product was previously available from the NASA National Space Science Data Center (NSSDC) 

under the name TIROS 3 Final Meteorological Radiation Files (FMRT) with the identifier ESAD-00141 

(old id 61-017A-03A).

1.1.1 Medium Resolution Scanning Radiometer

The scanning radiometer on the TIROS 3 meteorological satellite measured the emitted and 

reflected radiation of the earth and its atmosphere. The five-channel radiometer scanned the earth 

and space about the satellite spin axis. The radiometer's bi-directional optical axes were inclined to 

the satellite spin axis at angles of 45° and 135°. The viewing directions are designated “wall” and 

“floor” according to their orientation in the satellite. When one optic views the earth, the other 

views outer space. As the satellite rotates on. its axis, the radiometer scan pattern on the surface of 

the earth is defined by the intersection of a 45° half-angle cone and a sphere. This pattern ranges 

from a circle when the spin vector is parallel to the orbital radius vector, to two hyperbola-like 

branches when the spin vector is perpendicular to the radius vector. When viewing downward from 

a height of 750 km, the spatial resolution of the radiometer is about 65 km.

The five-channel sensor used bolometer detectors and filters to limit the spectral response and to 

provide comprehensive data by measuring radiation intensities in selected portions of the infrared 

spectrum. The spatial bandwidth of each channel (in microns) and its associated parameter were as 

follows:

Channel 1 – 6.0 to 6.5 (water vapor absorption)

Channel 2 – 8.0 to 12.0 (atmospheric window)

Channel 3 – 0.25 to 6.0 (reflected solar radiation)

Channel 4 – 8.0 to 30.0 (terrestrial radiation)

Channel 5 – 0.55 to .075 (reflected solar radiation)

Response characteristics of all channels degraded rapidly after launch. The greatest uncertainty in 

the radiation measurements is due to the apparent shift in the zero radiation level. Data are usable 

for channels 1, 2, 3, 4, and 5 up to orbits 118, 875, 875, 130, and 300, respectively.

Data are available for the two and a half month time period from July 12, 1961 to September 30, 

1961. A previous version of this instrument was flown on TIROS 2, and subsequent versions were 

flown on the TIROS 4 and 7 satellites.

1.1.2 TIROS 3 Overview

The second Television and InfraRed Observation Satellite (TIROS 3) was launched into orbit on July
12,  1961.  TIROS 3 was a spin-stabilized meteorological  spacecraft designed to test  experimental
television techniques and infrared equipment. The satellite was in the form of an 18-sided right
prism, 107 cm in diameter and 56 cm high. The top and sides of the spacecraft were covered with
approximately  9000  1-cm  by  2-cm  silicon  solar  cells.  It  was  equipped  with  two  independent
television  camera  subsystems  for  taking  cloudcover  pictures,  plus  a  two-channel  low-resolution
radiometer,  an  omnidirectional  radiometer,  and  a  five-channel  infrared  scanning  radiometer.  All
three  radiometers  were  used  for  measuring  radiation  from  the  earth  and  its  atmosphere.  The
satellite spin rate was maintained between 8 and 12 rpm by use of five diametrically opposed pairs
of small, solid-fuel thrusters. A magnetic attitude control device permitted the satellite spin axis to
be oriented to within 1 to 2 deg of a predetermined attitude. The spacecraft performed normally
until  August  1961,  when the scanning radiometer began to fail.  Performance was sporadic  until
January 23, 1962. It was deactivated on February 28, 1962.

The orbit of the satellite can be characterized by the following:

 Perigee Altitude: 742 km

 Apogee Altitude: 812 km

 Orbital Period: 100.41 minutes

 Inclination: 47.90 degrees

 Eccentricity: 0.00489

1.2 Algorithm Background

The TIROS 3 FMRT data were generated from the spacecraft telemetry, attitude and orbital data . The
data were originally processed on IBM 7094 computers, and subsequently copied to 200 bpi 7-track
tapes for archival.   More detailed information on the TIROS 3 scanning radiometer instrument and
data processing can be found in the TIROS III Radiation Data Users' Manual.

1.3 Data Disclaimer

The data should be used with care and one should first read the TIROS III Radiation Data Users’ 

Manual describing the scanning radiometer experiment and FMRT data. Users should cite this data 

product in their research:

Barksdale, Joseph D. (2021), TIROS3 Level 1 FMRT Data V001, Greenbelt, MD, USA, Goddard Earth 

Sciences Data and Information Services Center (GES DISC), Accessed: [Data Access Date], 

https://doi.org/10.5067/XQV4UWX1XJ8Q

2. Data Organization

The TIROS-3 Level 1 FMRT data spans the time period from July 12, 1961 to September 30, 1961. 

Each file contains one orbit of data.

2.1 File Naming Convention

The data product files are named according to the following convention:

<Platform>_<Level>-<Type>_<DateTime>_<Orbit>_<TapeNumber>-<FileNumber>.<Suffix>, where

o Platform = name of the platform or satellite (TIROS3)
o) Level = process level (L1)
o) Type =- Data type of data is Final Meteorlogical Radiation Tape (FMRT)
o) Date = Data start date in format <YYYY>m<MMDD> where

1. YYYY = 4 digit year (1961)
2. MM = 2 digit month (01 - 12)
3. DD = 2 digit day of month (01 - 31)
4. hh = 2 digit hour of day (00 - 23)
5. mm = 2 digit minute of hour (00 – 59)
6. ss = 2 digit second of minute (00 – 59)

o OrbitNumber = 5 digit orbit number preceded by letter ’o’
o TapeNumber = 5 digit number of tape (preceded by ‘DR’ - primary or ‘DS’ - backup)
o FileNumber = 3 digit number of file on tape
o Suffix = the file format (always TAP, indicating tape binary data)

File name example: TIROS3_L1-FMRT_1961m0829t192258_o00695_DS1054-210.TAP

2.2 File Format and Structure

The data are stored as they were originally written in IBM binary (big-endian) record oriented 
structured files. The files were written on 6250 bpi 9-track tapes or 1600 bpi 7-track tapes using a 
blocked FORTRAN format. Each tape has about 100 to 200 files on it, with each file containing one 
orbit of data. Each data file on the tape contains a set of data records with a FORTRAN record size 
word, the record block, and a FORTRAN record trailing size word.

The first record is a header or documentation record of size 64 bytes (9-track file) or 84 bytes (7-
track file). This is followed by a series of data records with variable byte sizes, depending on how 
many swaths are stored in a record. Files end with a single End-of-File word, the last file on the tape 
is followed by a double End-of-File word. Each data record will contain a set of swaths broken into 
several blocks. These blocks will contain five measurements for each of the five channels, as well as 
geolocation and other information. Latitude and longitude information is only provided for the first 
of each five measurements, the other four must be calculated. The data are written in big-endian 

IBM 36-bit integers, stored either in 4 ½ 8 bytes (9-track file) or in 6 bytes (7-track file) where only 
the first 6 bits are used. For the contents and layout of the data, see section 3.1 below.

During data recovery there were two sets of tapes, The first set of tapes are the primary tapes and 
designated with a DR with a total of 2 tapes containing 212 files. The second set are the backup 
tapes which are designated with a DS also with a total of 2 tapes and 212 files. Each tape typically 
holds about 100 orbit files. During recovery the vast majority of DR and DS tape files were exact 
duplicates of each other, except three DR tape files were nearly identical with their DS counterparts 
but with extra, missing or corrupted data records. In the end there were 212 unique files from the 
secondary DS tapes which represent the complete record of the TIROS-3 Level 1 FMRT data 
collection and are publicly available from the GES DISC.

2.3 Key Science Data Fields

The primary science data fields are radiances expressed as effective radiant emmitance (W/m2).

Figure 1: Typical data coverage for a TIROS-3 Level 1 FMRT orbit file, red line shows subsatellite track

Figure 2: Scanning modes of the TIROS radiometer: (a) closed mode – circular, (b) single open mode 

– parabola, and (c) alternating open mode – hyperbola

3. Data Contents

The granularity for the FMRT data is one orbit.

3.1 Data Records

The file format of these data is described in the document “TIROS VII Radiation Data Catalog and 

User’s Manual, Volume 1". A summary of this is described in the tables below.

Each of the original tapes contain data files that have  one orbit of data, with each day containing 

~14 orbits. Each orbit file begins with a 64-byte (9-track) or an 84-byte (7-track) documentation or 

header record, which is followed by a series of data records of variable length. Each data record 

contains a set of swaths which are made up of multiple blocks with 5 measurements for each 

channel. The data were written as IBM big-endian 36-bit integer words and now stored in 4 ½ 8 

bytes (9-track) or 6 bytes (7-track) with only first 6 bits used. Data record values are split into 2 half 

words, the decrement (WordD) and the address (WordA) of 18-bits each . There are a total of 677 

unique data files for the entire mission spanning the 5 month time period from July 12, 1961 to 

September 30, 1961.

Table 3-1: FMRT Documentation Record

Word Quantity Units Scaling Remarks

1 Dref - 1 0 = 1 September 1957

2 Date - 1 octal (mmddyy)

3 Start Day - 1 Days since Dref

4 Start Hour - 1

5 Start Minute - 1

6 Start Seconds - 29

7 End Day - 1

8 End Hour - 1

9 End Minute - 1

10 End Seconds - 29

11 Satellite Spin Rate deg/sec 29 25x4x14

12 Data Sampling Frequency 36,72,144 1 (cycles of a 550 cps tuning fork)

13 Orbit Number - 1

14 Station Code - 1
1 = Wallops Island, Virginia

2 = San Nicolas Island, California

Table 3-2: FMRT Data Record

Word Function Units Scaling Remarks

1A Day - 1

1D Hour - 1

2D Minute - 1

2A Greenwich hour angle (GHA) deg 26 at time specified in words 1D, 1A, and 2D

3D Declination deg 26
at time specified in words 1D, 1A, and 2D, 

90° added to be positive

3A Reference temperature of radiometer Kelvin 1

4D Reference temperature of electronics Kelvin 1

4A Height of spacecraft km 1 at time specified in words 1D, 1A, and 2D

5D Latitude of initial subsatellite point deg 26 at time specified in words 1D, 1A, and 2D, 

90° added to be positive

5A Longitude of initial subsatellite point deg 26
at time specified in words 1D, 1A, and 2D, 

0 – 360° with west being positive

6D Seconds - 29 at time specified in words 1D, 1A, and 2D

6A Latitude of subsatellite point deg 26 at time specified in words 1D, 1A, 2D and 6D, 

90° added to be positive

7D Longitude of subsatellite point deg 26
at time specified in words 1D, 1A, 2D and 6D, 

0 – 360° with west being positive

7A Latitude of viewed point on earth deg 26 at time specified in words 1D, 1A, 2D and 6D, 

90° added to be positive

8D Longitude of viewed point on earth deg 26 at time specified in words 1D, 1A, 2D and 6D, 

0 – 360° with west being positive

8A
Nadir angle of optic axis from 
radiometer to point in 7A, 8D

deg 26

9D
Azimuth angle of optic axis from 
radiometer to point in 7A, 8D

deg 26 measured clockwise from north and 

expressed as a positive number

9A Zero - - at time specified in words 1D, 1A, 2D and 6D

10D Ch. 1 TBB for first measurement K 23 at time specified in words 1D, 1A, 2D and 6D

10A Ch. 2 TBB for first measurement K 23 at time specified in words 1D, 1A, 2D and 6D

11D Ch. 3 W for first measurement W/m2 23 at time specified in words 1D, 1A, 2D and 6D

11A Ch. 4 TBB for first measurement K 23 at time specified in words 1D, 1A, 2D and 6D

12D Ch. 5 W for first measurement W/m2 23 at time specified in words 1D, 1A, 2D and 6D

12A Zero

13D

24A
Measurements for times 2, 3, 4 and 5

words 10,11, and 12 repeat four times 

following first measurement (five total)

The block of data from words 6D to 24A will be repeated, thus defining every fifth measurement in a swath.

The address (WordA) of the third to last word of a swath will contain ‘010101010101010’b to signal end-of-record.

Each end-of-swath is followed by the following two words:

(N)D Code indicating end of swath - - ‘111111111111111’b

(N)A Minimum nadir angle of swath deg 26

(N+1)D Latitude of point of min. nadir angle deg 26 90° added to be positive

(N+1)A Longitude of point of min. nadir angle deg 26 0 – 360° with west being positive

3.2 Metadata

The metadata are contained in a separate XML formatted file having the same name as the data file 

with .xml appended to it.

Table 3-2: Metadata attributes associated with the data file.

Name Description

LongName Long name of the data product.

ShortName Short name of the data product.

VersionID Product or collection version.

GranuleID Granule identifier, i.e. the name of the file.

Format File format of the data file.

CheckSumType Type of checksum used.

CheckSumValue The value of the calculated checksum.

SizeBytesDataGranule Size of the file or granule in bytes.

InsertDateTime Date and time when the granule was inserted into the archive.  The format for date is YYYY-

MM-DD and time is hh-mm-ss.

ProductionDateTime Date and time the file was produced in format YYYY-MM-DDThh:mm:ss.ssssssZ

RangeBeginningDate Begin date when the data was collected in YYYY-MM-DD format.

RangeBeginningTime Begin time of the date when the data was collected in hh-mm-ss format.

RangeEndingDate End date when the data was collected in YYYY-MM-DD format.

RangeEndingTime End time of the date when the data was collected in hh-mm-ss format.

PlatformShortName Short name or acronym of the platform or satellite

InstrumentShortName Short name or acronym of the instrument 

SensorShortName Short name or acronym of the sensor

GPolygon: 
PointLatitude

Latitudes of the polygon (rectangle) points that represent the satellite coverage. Each point is
identified by its latitude and longitude pair.

GPolygon: 
PointLongitude

Longitudes of the polygon (rectangle) points that represent the satellite coverage. Each point 
is identified by its latitude and longitude pair.

Orbit Satellite orbit number

Station Code Ground station where the data was downlinked

ElapsedMinTime Duration in minutes of data collected.

4. Reading the Data

The data are written in a binary record-oriented format. Using the record format specification in the 

section above, users can write software to read the data files. Please note that the data were 

originally written as IBM 36-bit words in big-endian order, therefore users will need to read in 4 ½ 

bytes (9-track) or 6 bytes (7-track) with only 6 bits from each are used, to reconstruct each of the 

original 36-bit words. The 36-bit words in the data records are split into two half words, the 

decrement (wordD) and address (wordA). If second to last wordD = ‘111111111111111’b this 

indicates end-of-swath. If third to last wordA = ‘010101010101010’b this indicates end-of-record.

A sample FORTRAN program is included in the Appendix section which will read and print the the 

data contents, with functions to convert the data stored in their original IBM 36-bit words.

Table 4-1: Orbit ranges for tape files

9-track files 7-track files

1 – 5 15 – 134

142 – 160 171 – 246

256 – 261 270 – 273

282 – 288 311 – 369

381 – 384 397 – 427

438 – 600 607 – 614

612 – 897 907

939 – 940 950

964 – 968 977 – 981

991 1023 – 1037

1047 – 1053 1088 – 1094

1103 – 1152

5. Data Services

5.1 GES DISC Search

The GES DISC provides a keyword, spatial, temporal and advanced (event) searches through its 

unified search and download interface:

https://disc.gsfc.nasa.gov/

5.2 Documentation

The data product landing pages provide information about these data products, as well as links to 
download the data files and relevant documentation:

https://disc.gsfc.nasa.gov/datacollection/TIROS3L1FMRT_001.html

5.3 Direct Download

These data products are available for users to download directly using HTTPS:

https://acdisc.gesdisc.eosdis.nasa.gov/data/TIROS/TIROS3L1FMRT.001/

6. More Information

6.1 Contact Information

Name: GES DISC Help Desk

URL: https://disc.gsfc.nasa.gov/

E-mail: gsfc-help-disc@lists.nasa.gov

Phone: 301-614-5224

Fax: 301-614-5228

Address: Goddard Earth Sciences Data and Information Services Center

Attn: Help Desk

Code 610.2

NASA Goddard Space Flight Center

Greenbelt, MD  20771, USA

6.2 References

“TIROS III Radiation Data Users’ Manual”, NASA Goddard Space Flight Center, August 1962

“TIROS III Radiation Data Catalog”, NASA Goddard Space Flight Center and the US Weather Bureau, 

December 15 1962

Acronyms

EOS: Earth Observing System

ESDIS: Earth Science and Data Information System

GES DISC: Goddard Earth Sciences Data and Information Services Center

GSFC: Goddard Space Flight Center

L1: Level-1 Data

NASA: National Aeronautics and Space Administration

TIROS: Television Infrared Observation Satellite

QA: Quality Assessment

UT: Universal Time

FORTRAN Code
C-----------------------------------------------------------------------
C ^NAME: READ_FMRT
C   This program reads a TIROS Scanning Radiometer L1 FMRT data file.
C   The TIROS FMRT files contain a documentation or header record
C   (size 64 bytes 9-track tape or 84 bytes 7-track tape), followed
C   byt a series of data records (of varying length). The data records
C   contain a set of swath data broken into several blocks, each block
C   containing typically five measurements from each channel of the
C   scanning radiometer. The data were originally written as big-endian
C   ordered IBM 36-bit words which are now packed into 4-1/2 bytes
C   (9-track tapes) or 6 bytes with each only using 6-bits (7-track
C   tapes). The documentation record contains fourteen 36-bit words.
C   The words in the data records are split into two halves, referred to
C   as the decrement (Word-D) and the address (Word-A). When the second
C   to last Word-D equals '111111111111111'b, this designates the end of
C   a swath. When the third to last Word-A is '010101010101010'b this
C   indicates the end of a data record. Each measurment may be viewed
C   from the floor or wall of the spacecraft. A wall view is designated
C   by bit 19 of the word measurement pairs Ch. 1/2 and Ch. 3/4. If 
C   bit 18 of the word for measurment pair Ch. 3/4 is set this indicates
C   visible channels 3 and 5 are saturated. A negative word (bit 35) is
C   typically marks the first and last swath when the instumrnet is in
C   the closed mode, or some other problem. This program will print the
C   contents of each data record. Please see the TIROS Radation Data
C   Catalog and User's Guide for more detailed information.
C
C ^MAJOR VARIABLES:
C     FNAME  - name of input file
C     IBLKSZ - size of record block in bytes
C     BUFF   - buffer for data record
C     WD36   - array for 36-bit words
C     IOS    - I/O status number
C
C ^NOTES:
C     Compile: gfortran -o READ_FMRT.EXE READ_FMRT.FOR
C
C ^AUTHOR:  James Johnson (James.Johnson@nasa.gov), NASA GES DISC
C
C ^HISTORY: Feb 27, 2021 - first version
C           May  3, 2021 - support for both 7- and 9-track tape words
C-----------------------------------------------------------------------
      PROGRAM READ_FMRT
      CHARACTER FNAME*1024                   ! Filename
      CHARACTER BUFF(18000)                  ! Buffer for data record block
      INTEGER*8 WD36(4000)                   ! Array for 36-bit words
      INTEGER*4 IBLKSZ                       ! Size of records
      INTEGER*4 IWORD                        ! 4-byte word
      REAL*4 NPACK                           ! 36-bit word packing method
      CHARACTER*1 RECFLG                     ! Record parity flag
      CHARACTER TEMP(4)                      ! Buffer to hold 4-byte word
      EQUIVALENCE (TEMP,IWORD)

C     Get the name of the input data file to read
      WRITE (0, *), 'Enter the name of the input file:'
      READ (5,'(A)') FNAME
      PRINT '("File = ",A160)', FNAME

C     Open the specified input file
      OPEN (UNIT=1, FILE=FNAME, STATUS='OLD', ACCESS='DIRECT',
     &      FORM='UNFORMATTED', RECL=1, ERR=99, IOSTAT=IOS)

C     Initialize N (record number) and IOFF (byte offset in file)
      N=0
      IOFF=0

C     Loop through the file reading all records in file
      DO

C       Read the first 4-byte word or record size header
        DO I=1,4
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) TEMP(I)
          IBLKSZ = IWORD
        END DO
        IOFF=IOFF+(I-1)
        IF (ISHFT(IWORD,-31) .EQ. 1) THEN                  ! Check Bit 31
          IBLKSZ = IAND(IWORD,'7FFFFFFF'Z)
          RECFLG = "*"
        ELSE
          RECFLG = " "
        END IF

C       End-of-File (EOF) mark, break out of do loop
        IF (IBLKSZ .EQ. 0) GOTO 90

C       Next read the block of data
        DO I=1,11250
          BUFF(I)=CHAR(0)                                  ! Clear the buffer
        END DO
        DO I=1,IBLKSZ
          READ (1, REC=IOFF+I, IOSTAT=IOS) BUFF(I)
          IF (IOS .NE. 0) THEN
            PRINT '("ERROR: BUFF ",I4,X,I4,", IOSTAT: ",I6)', N,I-1,IOS
            IBLKSZ = I-1
            GOTO 90
          END IF
        END DO
        IOFF=IOFF+(I-1)
        PRINT '("=====================")'
        IF (RECFLG .EQ. "*") THEN
          PRINT '("RECORD: ",I3,A1)', N, RECFLG
        ELSE
          PRINT '("RECORD: ",I3)', N
        END IF
        PRINT '("=====================")'
        N=N+1

        IF (N .EQ. 1) THEN
          IF (IBLKSZ .EQ. 84) THEN
            NPACK = 6.0                                    ! 9-track tape

          ELSE
            NPACK = 4.5                                    ! 7-track tape
          END IF
        END IF

        CALL BLK2WD(BUFF,IBLKSZ,WD36,NWORDS,NPACK)

        IF (N .EQ. 1 .AND. IBLKSZ .EQ. 64 .OR. IBLKSZ .EQ. 84) THEN
          CALL PRHEAD(NWORDS,WD36(1:NWORDS))               ! Doc/Header Record
        ELSE
          CALL PRDREC(NWORDS,WD36)                         ! Data Record
        END IF

        DO I=1,4                                           ! EOR word rec size
          READ (1, REC=IOFF+I, IOSTAT=IOS, ERR=90) TEMP(I)
        END DO
        IOFF=IOFF+(I-1)

        IF (IBLKSZ .NE. IWORD) THEN
          PRINT '("WARN: REC SIZE MISMATCH ",I4,X,I4)', IBLKSZ, IWORD
        END IF

      END DO

C     Close the input file
   90 CLOSE(1)
      STOP

   99 PRINT '("ERROR: OPEN FILE, IOSTAT: ",I6)', IOS
      STOP
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the Header/Documentation Record
C-----------------------------------------------------------------------
      SUBROUTINE PRHEAD(NWORDS,WD36)
      INTEGER*8 WD36(14)                     ! Array for 36-bit words

      PRINT '("Dref       =",X,I8)', WD36(1)
      PRINT '("Int. Date  =",X,O8.6)', WD36(2)
      PRINT '("Start Day  =",X,I8)', WD36(3)
      PRINT '("Start Hour =",X,I8)', WD36(4)
      PRINT '("Start Min  =",X,I8)', WD36(5)
      PRINT '("Start Sec  =",X,F8.3)', WD36(6)/2.**(35-26)
      PRINT '("End Day    =",X,I8)', WD36(7)
      PRINT '("End Hour   =",X,I8)', WD36(8)
      PRINT '("End Min    =",X,I8)', WD36(9)
      PRINT '("End Sec    =",X,F8.3)', WD36(10)/2.**(35-26)
      PRINT '("Spin Rate  =",X,F8.3)', WD36(11)/2.**(35-26)
      PRINT '("Frequency  =",X,I8)', WD36(12)
      PRINT '("Orbit      =",X,I8)', WD36(13)
      PRINT '("Station    =",X,I8)', WD36(14)

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine will print the Data Record
C-----------------------------------------------------------------------
      SUBROUTINE PRDREC(NWORDS,WD36)
      INTEGER*8 WD36(2500)                           ! Data record(36-bit words)
      INTEGER*4 WORDD                                ! Word decrement (18-bits)
      INTEGER*4 WORDA                                ! Word address (18-bits)
      REAL*4 CHANNEL(5,5)                            ! Array for channel data
      INTEGER*2 ZEROS(5)                             ! Array for spare zeros
      CHARACTER*1 CHFLAG(5,2)                        ! Array for wall flag
      CHARACTER*64 SFMT                              ! Format statement
      
      ISW = 1                                        ! Swath counter
      I = 0                                          ! Swath word counter
      IX = 0                                         ! End-Of-Swath word counter
      IERC = 0                                       ! End-Of-Record (false)
      IEOS = 0                                       ! End-Of-Swath flag (false)

C     Data Record
      DO N=1,NWORDS

C       Bit-35 negative flag (measurement is begin/end of swath closed mode)
        IF (ISHFT(WD36(N),-35) .EQ. 1) THEN
          ISGN =-1
        ELSE
          ISGN = 1
        END IF

        WORDD = ISHFT(WD36(N),-18)
        WORDA = IAND(WD36(N),'3FFFF'Z)

        IF (IAND(WORDD,'7FFF'Z) .EQ. '111111111111111'B) THEN
          IEOS=1
        END IF
        IF (IAND(WORDA,'7FFF'Z) .EQ. '010101010101010'B) THEN
          IERC=1
        END IF

        I = I+1                                      ! Word in swath
        J = MOD((I-6),19)+6                          ! Word in block
        IBLK = (I-6)/19+1                            ! Block number
        IF (J .GE. 10) THEN
          IF (IEOS .NE. 1) THEN
            M = (J-10)/3+1                           ! Measurement number
          END IF
        ELSE
          M = 0
        END IF

c       PRINT'("N:",I4," I:",I3,", ISW:",I3,", IBLK:",I3,", J:",I3,
c    +  ", M:",I3,", EOS:",I3)', N,I,ISW,IBLK,J,M,IEOS

        IF (IEOS .EQ. 0) THEN                        ! End-of-Swath check
          IF      (J .EQ. 1) THEN
            PRINT '("Day        =",X,I8)', WORDD
            PRINT '("Hour       =",X,I8)', WORDA
          ELSE IF (J .EQ. 2) THEN

            PRINT '("Minute     =",X,I8)', WORDD
            PRINT '("GHA        =",X,F8.3)', WORDA/2.**(35-29)
          ELSE IF (J .EQ. 3) THEN
            PRINT '("Decl.      =",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("RefTempR   =",X,I8)', WORDA
          ELSE IF (J .EQ. 4) THEN
            PRINT '("RefTempE   =",X,I8)', WORDD
            PRINT '("Height     =",X,I8)', WORDA
          ELSE IF (J .EQ. 5) THEN
            PRINT '("LatSubSat0 =",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("LonSubSat0 =",X,F8.3)', WORDA/2.**(35-29)
          ELSE IF (J .EQ. 6) THEN
            IF (I .EQ. 6) THEN
              PRINT '("~~~~~~~~~~~~~~~~~~~~~")'
              PRINT '("SWATH:  ",I3)', ISW
              PRINT '("~~~~~~~~~~~~~~~~~~~~~")'
            END IF
            PRINT '("BLOCK:  ",I3)', IBLK
            PRINT '("---------------------")'
            PRINT '("Seconds    =",X,F8.3)', WORDD/2.**(17-8)
            PRINT '("LatSubSat  =",X,F8.3)', WORDA/2.**(35-29)
          ELSE IF (J .EQ. 7) THEN
            PRINT '("LonSubSat  =",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("Latitude   =",X,F8.3)', WORDA/2.**(35-29)
          ELSE IF (J .EQ. 8) THEN
            PRINT '("Longitude  =",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("Nadir      =",X,F8.3)', WORDA/2.**(35-29)
          ELSE IF (J .EQ. 9) THEN
            PRINT '("Azimuth    =",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("Zero       =",X,I8)', WORDA
          ELSE IF (J .EQ. 10 .OR. J .EQ. 13 .OR.
     +             J .EQ. 16 .OR. J .EQ. 19 .OR. J .EQ. 22) THEN
            CHANNEL(M,1) = ISGN*IAND(WORDD,'FFFF'Z)/2.**(17-14)
            CHANNEL(M,2) = ISGN*IAND(WORDA,'FFFF'Z)/2.**(35-32)
          ELSE IF (J .EQ. 11 .OR. J .EQ. 14 .OR.
     +             J .EQ. 17 .OR. J .EQ. 20 .OR. J .EQ. 23) THEN
            CHANNEL(M,3) = ISGN*IAND(WORDD,'FFFF'Z)/2.**(17-14)
            CHANNEL(M,4) = ISGN*IAND(WORDA,'FFFF'Z)/2.**(35-32)
            IF (IAND(ISHFT(WD36(N),19-35),'01'Z) .EQ. 1) THEN
              CHFLAG(M,1) = "W"                      ! Wall/floor view flag
            ELSE
              CHFLAG(M,1) = "F"
            END IF
            IF (IAND(ISHFT(WD36(N),18-35),'01'Z) .EQ. 1) THEN
              CHFLAG(M,2) = "S"                      ! Ch.3/5 saturation flag
            ELSE
              CHFLAG(M,2) = "_"
            END IF
          ELSE IF (J .EQ. 12 .OR. J .EQ. 15 .OR.
     +             J .EQ. 18 .OR. J .EQ. 21 .OR. J .EQ. 24) THEN
            CHANNEL(M,5) = ISGN*IAND(WORDD,'FFFF'Z)/2.**(17-14)
            ZEROS(M) = WORDA
            IF (J .EQ. 24) THEN
              PRINT '("Channel 1  =",5(X,F8.3),5(2X,A1))',
     +          CHANNEL(1:5,1), CHFLAG(1:5,1)
              PRINT '("Channel 2  =",5(X,F8.3),5(2X,A1))',
     +          CHANNEL(1:5,2), CHFLAG(1:5,1)

              PRINT '("Channel 3  =",5(X,F8.3),5(2X,A1))',
     +          CHANNEL(1:5,3), CHFLAG(1:5,2)
              PRINT '("Channel 4  =",5(X,F8.3),5(2X,A1))',
     +          CHANNEL(1:5,4), CHFLAG(1:5,1)
              PRINT '("Channel 5  =",5(X,F8.3),5(2X,A1))',
     +          CHANNEL(1:5,5), CHFLAG(1:5,2)
              PRINT '("Zeros      =",5(X,I8))', ZEROS(1:5)
              PRINT '("---------------------")'
            END IF
          END IF
        ELSE                                         ! End-Of-Swath found
          IF (IX .EQ. 0) THEN
            IF (M .NE. 0) THEN                       ! Incomplete block
              WRITE(SFMT, "(A,I1,A,I1,A)")
     +          '("Channel 1  =",', M, '(X,F8.3),', M, '(2X,A1))'
              PRINT(SFMT), CHANNEL(1:M,1), CHFLAG(1:M,1)
              WRITE(SFMT, "(A,I1,A,I1,A)")
     +          '("Channel 2  =",', M, '(X,F8.3),', M, '(2X,A1))'
              PRINT(SFMT), CHANNEL(1:M,2), CHFLAG(1:M,1)
              WRITE(SFMT, "(A,I1,A,I1,A)")
     +          '("Channel 3  =",', M, '(X,F8.3),', M, '(2X,A1))'
              PRINT(SFMT), CHANNEL(1:M,3), CHFLAG(1:M,2)
              WRITE(SFMT, "(A,I1,A,I1,A)")
     +          '("Channel 4  =",', M, '(X,F8.3),', M, '(2X,A1))'
              PRINT(SFMT), CHANNEL(1:M,4), CHFLAG(1:M,1)
              WRITE(SFMT, "(A,I1,A,I1,A)")
     +          '("Channel 5  =",', M, '(X,F8.3),', M, '(2X,A1))'
              PRINT(SFMT), CHANNEL(1:M,5), CHFLAG(1:M,2)
              WRITE(SFMT, "(A,I1,A)")
     +          '("Zeros      =",', M, '(X,I8))'
              PRINT(SFMT), ZEROS(1:M)
              PRINT '("---------------------")'
            END IF

            PRINT '("Code       =",X,B15.15)', WORDD
            PRINT '("MinNadir   =",X,F8.3)', WORDA/2.**(35-29)
            IX = IX+1
          ELSE
            PRINT '("LatMinNadir=",X,F8.3)', WORDD/2.**(17-11)
            PRINT '("LonMinNadir=",X,F8.3)', WORDA/2.**(35-29)
            IEOS = 0
            I = 5     ! Reset I=5 because words 1-5 only for beginning of record
            IX = 0
            ISW = ISW+1
          END IF
        END IF
      END DO

      RETURN
      END

C-----------------------------------------------------------------------
C     This Subroutine unpacks 36-bit words into 64-bit long long integer
C-----------------------------------------------------------------------
      SUBROUTINE BLK2WD(BUFF, IBLKSZ, WD36, NWORDS, NPACK)
      CHARACTER BUFF(11250)                          ! Buffer for record block
      REAL*4 NPACK                                   ! 36-bit packing method
      INTEGER*8 W36BIT                               ! Use 64-bit for 36-bit int
      INTEGER*8 WD36(2500)                           ! Array of 64-bit words

C     Store 36-bit words into 64-bit double long IEEE integers
      NWORDS = INT(IBLKSZ/NPACK)
      DO N=1,NWORDS
        IF (NPACK.EQ.4.5) THEN
          WD36(N) =
     &    W36BIT(BUFF(INT((N-1)*NPACK)+1:INT(N*NPACK+0.5)), N)
        ELSE
          WD36(N) =
     &    W36BIT(BUFF(INT((N-1)*NPACK)+1:INT(N*NPACK)), 0)
        END IF
      END DO

      RETURN
      END

C-----------------------------------------------------------------------
C     This Function will convert data into a 36-bit word
C-----------------------------------------------------------------------
      INTEGER*8 FUNCTION W36BIT(BUFF,N)
      CHARACTER BUFF(8)
      CHARACTER TEMP(8)
      INTEGER*8 WD36
      EQUIVALENCE (TEMP,WD36)

      IF (N.NE.0) THEN
        DO I=1,8
          IF (I.LE.5) THEN
            TEMP(8-I-2) = BUFF(I)                          ! swap the byte order
          ELSE
            TEMP(I) = CHAR(0)
          END IF
        END DO
        IF (MOD(N,2).EQ.0) THEN
          WD36 = IAND(WD36,'FFFFFFFFF'Z)
        ELSE
          WD36 = ISHFT(WD36,-4)
        END IF
      ELSE
        WD36 = 0
        DO I=1,6
          WD36 = ISHFT(WD36, 6)                            ! shift left 6 bits
          WD36 = IOR(WD36,IAND(ICHAR(BUFF(I)),'3F'Z))      ! remove 2 MSBs
        END DO
      END IF
      W36BIT = WD36

      RETURN
      END

